# LIKELIHOOD RATIO TESTS FOR REGRESSION COEFFICIENTS FROM THE COX PROPORTIONAL HAZARDS MODEL USING THE PHREG PROCEDURE

Laurent M. Kassalow, Sandoz Research Institute

## ABSTRACT

The PHREG procedure provides a TEST statement for testing hypotheses for regression coefficients from the Cox proportional hazards model. PHREG presents Wald tests for the test statistics. Many authors prefer likelihood ratio test statistics for testing these hypotheses. A SAS® macro will be presented that performs likelihood ratio tests for these hypothesis tests. This macro can also be used to perform stratified analyses.

## INTRODUCTION

The Cox proportional hazards model is a popular method for regression analysis of survival data. The estimates of the regression coefficients in the model and their standard errors may be obtained using the method of maximum likelihood. Three hypothesis testing methods for these coefficients based on large-sample theory are the likelihood ratio test, the Wald test, and the Score test, with the first two being the most commonly cited. These tests are described in detail in Cox and Oakes (1984), Lee, Harrell, Tolley, and Rosati (1983), Simon (1983), Kalbfleisch and Prentice (1980), and a more applied discussion may be found in Hosmer and Lemeshow (1989).

Many authors recommend the likelihood ratio test for hypothesis testing when using maximum likelihood estimation [Cox and Oakes (1984), Kalbfleisch and Prentice (1980), Hauck and Donner (1987), and Hosmer and Lemeshow (1989)]. SAS, via the PHREG procedure, provides a likelihood ratio test, Score test, and Wald test for testing the hypothesis that all regression coefficients are equal to zero ('Global' null hypothesis.) If one wishes to carry out hypothesis tests involving the significance of one regression coefficient or subset of regression coefficients, PHREG via its TEST statement provides only Wald statistics. Likelihood ratio tests can be performed for these tests using PHREG, however this involves running the PHREG procedure twice and deriving the desired likelihood ratio test statistic. A SAS macro is presented that performs this task.

## THE LIKELIHOOD RATIO TEST

The likelihood ratio test can be viewed in the context of a general test of a linear statistical model. Using the definitions and notation presented in Simon (1983), consider two models: the first contains p regression coefficients

(reduced or restricted model) and the second contains an additional q regression coefficients (full or unrestricted model). Let $\hat{\beta}_p$ and $\hat{\beta}_{p+q}$ denote the maximum likelihood estimates for the two models and $L_p$ and $L_{p+q}$ denote the values of the likelihood for the two models evaluated at $\hat{\beta}_p$ and $\hat{\beta}_{p+q}$, respectively. For large samples, the quantity

(Equation 1)

$$\Gamma_q=-2\log(L_p/L_{p+q})$$

has approximately a chi-square distribution with q degrees of freedom, under the null hypothesis that the additional q regression coefficients are all equal to zero.

The likelihood ratio test may be presented in a number of algebraically equivalent forms. The PHREG procedure, through the OUTEST statement, provides a SAS data set that contains the computed -2log(L) value. In order to use this quantity in a likelihood ratio test we must rearrange Equation 1 so that it is in a form representing a difference between two -2log(L) values. The following is the form of the equation that will be used:

(Equation 2)

$$\Gamma_q=-2\log(L_p)-[-2\log(L_{p+q})]$$

SAS MACRO

The macro is listed in the beginning of the SAS program in Appendix B. The macro assumes that the variable name for the survival time is STIME and the variable representing censoring status is called CEN. It is further assumed that CEN has the value of "1" for "Dead" (or any event being considered) and a value of "0" for "Censored" (or absence of any event being considered). The macro variables are coded to tell SAS the SAS data set name (DAT), the independent variables in the full model (FULLIND) and reduced model (REDIND), and the degrees of freedom of the two models, respectively (DFFULL and DFRED). The degrees of freedom is the number of independent variables in the model. COMP is the title of the hypothesis tested by the likelihood ratio test and is placed in TITLE4.

Three options are available. The two macro variables PRINTF and PRINTR control the printing of the individual runs of the PHREG procedure for the full and reduced models respectively.

STRATAV is coded with the value of a stratification variable if a stratified analysis is desired [Kalbfleisch and Prentice (1980) and Simon (1983)]. With a stratified model, the likelihood is the product of the individual stratum specific likelihoods. Stratification can be used to

adjust for factors that one chooses not to model, either to reduce the number of regression coefficients to estimate or to reduce perceived violations of the proportional hazards assumption. The stratified approach is essentially equivalent to a Mantel-Haenzel analysis [Simon (1983)].

The macro executes PROC PHREG twice (for the full and reduced models) and creates two SAS data sets using the OUTEST statement. The data sets are merged and the likelihood ratio test is computed based on Equation 2. The p-value corresponding to the computed chi-square and appropriate degrees of freedom is also computed. An example will illustrate the procedure.

## EXAMPLE

The example is from Anderson, Auquier, Hauck, et al. (1980), Pages 202-219. The example involves a comparison of remission times between a treatment and a control group among leukemia patients. The first hypothesis considered is whether there is an interaction between treatment group and the patient's initial white blood cell count. The logarithm of the count is the variable actually analyzed. The full model has independent variables for treatment group (TRT), logarithm of initial white blood cell count (LOGWBC), and a term representing the interaction of TRT and LOGWBC (TRT_LWBC). The latter is derived by multiplying TRT and LOGWBC for each patient. The reduced model has two independent variables, TRT and LOGWBC. The data analyzed and the input statement are listed in Appendix B. The appropriate macro call is shown in Appendix B, Example 1. The output produced is listed in Appendix A, Example 1.

Another macro call is presented (Appendix B, Example 2) that tests for significant between group differences after adjusting for LOGWBC. The interaction term is dropped from the full model for this analysis. The results are shown in Appendix A, Example 2.

The final macro call (Appendix B, Example 3) illustrates the use of a stratification variable. The hypothesis being tested is whether there is an interaction between treatment group and the patient's initial white blood cell count. The analysis is similar to Example 1, except that TRT is a stratification variable and is not entered into the model as an indicator variable. The results of the hypothesis test, shown in Appendix A, Example 3 are similar to those observed in Example 1.

## REFERENCES

Anderson, S., Auquier, A., Hauck, W.W, et al. (1980), *Statistical Methods for Comparative Studies: Techniques for Bias Reduction*, New York: John Wiley & Sons, Inc.

Cox, D.R. and Oakes, D. (1984), *Analysis of Survival Data*, London: Chapman and Hall.

Hauck, W.W., and Donner, A. (1977), "Wald's Test as Applied to Hypotheses in Logit Analysis," *Journal of the American Statistical Association*, **72**, 851-853.

Hosmer, D.W. and Lemeshow, S. (1989), *Applied Logistic Regression*, New York: John Wiley & Sons, Inc.

Kalbfleisch, J.D. and Prentice, R.L. (1980), *The Statistical Analysis of Failure Time Data*, New York: John Wiley & Sons, Inc.

Lee, K.L., Harrell, F.E., Tolley, H.D., and Rosati, R.A. (1983), "A Comparison of Test Statistics for Assessing the effects of Concomitant Variables in Survival Analysis," *Biometrics*, **39**, 541-350.

Simon, R. (1983), "Use of Regression Models: Statistical Aspects," in M.E. Buyse, M.J. Staquet, R.J. Sylvester (ed.), *Cancer Clinical Trials: Design, Practice and Analysis*, London: Oxford University Press.



## APPENDIX A*

EXAMPLE 1
TEST FOR TREATMENT-BY LOGWBC INTERACTION

| -2 LOG(L) FULL | -2 LOG(L) REDUCED | DEGREES OF FREEDOM | LIKELIHOOD RATIO CHI-SQUARE | P-VALUE |
|---|---|---|---|---|
| 144.131 | 144.559 | 1 | 0.427 | 0.5143 |

EXAMPLE 2
TEST FOR TREATMENT GROUP DIFFERENCES AFTER ADJUSTING FOR LOGWBC

| -2 LOG(L) FULL | -2 LOG(L) REDUCED | DEGREES OF FREEDOM | LIKELIHOOD RATIO CHI-SQUARE | P-VALUE |
|---|---|---|---|---|
| 144.559 | 154.946 | 1 | 10.387 | 0.0013 |

EXAMPLE 3
TEST FOR TREATMENT-BY-LOGWBC INTERACTION

| -2 LOG(L) FULL | -2 LOG(L) REDUCED | DEGREES OF FREEDOM | LIKELIHOOD RATIO CHI-SQUARE | P-VALUE |
|---|---|---|---|---|
| 114.888 | 115.250 | 1 | 0.362 | 0.5472 |

*ANDERSON, ET AL, COX PROPORTIONAL HAZARDS MODEL (PAGES 202-219)

## APPENDIX B

```
   /* LIKMACRO SAS */
*******************************************************;
   /*            DATA SET ASSUMPTIONS                  */
   /* CEN - CENSORING VARIABLE - 1='DEAD', 0='CENSORED' */
   /* STIME - SURVIVAL TIME                            */
   /*            MACRO VARIABLE DEFINITIONS            */
   /* DAT - INPUT SAS DATA SET                         */
   /* FULLIND - TERM(S) IN THE FULL MODEL              */
   /* DFFULL - DEGREES OF FREEDOM IN THE FULL MODEL
      (NUMBER OF TERMS IN THE FULL MODEL)              */
   /* REDIND - TERMS IN THE REDUCED MODEL              */
   /* DFRED - DEGREES OF FREEDOM IN THE REDUCED MODEL
      (NUMBER OF TERMS IN THE REDUCED MODEL)           */
   /* COMP - TITLE FOR COMPARISON
      PRINTED AS TITLE4                                */
   /* PRINTF - PRINT PHREG OUTPUT FOR FULL MODEL
      N FOR NO                                         */
   /* PRINTR - PRINT PHREG OUTPUT FOR REDUCED MODEL
      N FOR NO                                         */
   /* STRATAV - STRATIFICATION VARIABLE IF STRATIFIED
      ANALYSIS IS WANTED.  BLANK IF NOT WANTED         */
%MACRO LR(DAT=,FULLIND=,DFFULL=,REDIND=,DFRED=,
PRINTF=,PRINTR=,COMP=,STRATAV=);
PROC PHREG DATA=&DAT OUTEST=FULL
%IF &PRINTF=N %THEN %DO;
   NOPRINT
%END;
   ;
   MODEL STIME*CEN(0)=&FULLIND / ALPHA=0.05 RL;
%IF &STRATAV NE ' ' %THEN %DO;
   STRATA &STRATAV;
%END;
   TITLE4 "&COMP ";
   TITLE5 'FULL MODEL';
RUN;
PROC PHREG DATA=&DAT OUTEST=RED
%IF &PRINTR=N %THEN %DO;
   NOPRINT
%END;
   ;
   MODEL STIME*CEN(0)=&REDIND / ALPHA=0.05 RL;
%IF &STRATAV NE ' ' %THEN %DO;
   STRATA &STRATAV;
%END;
   TITLE4 "&COMP ";
   TITLE5 'REDUCED MODEL';
RUN;
DATA LRTEST;
   MERGE FULL(IN=INA RENAME=(_LNLIKE_=LFULL))
         RED(IN=INB RENAME=(_LNLIKE_=LRED));
   LIKFULL=-2*LFULL;
```

```
   LIKRED=-2*LRED;
   DFLR=&DFFULL-&DFRED;
   LRCHISQ=LIKRED-LIKFULL;
   PVAL=1-PROBCHI(LRCHISQ,DFLR);
   LABEL LIKFULL='-2 LOG(L) FULL'
         LIKRED='-2 LOG(L) REDUCED'
         DFLR='DEGREES OF FREEDOM'
         LRCHISQ='LIKELIHOOD RATIO CHI-SQUARE'
         PVAL=   'P-VALUE'
         ;
   KEEP LIKFULL LIKRED DFLR LRCHISQ PVAL;
   FORMAT LRCHISQ 10.3 PVAL 10.4;
RUN;
PROC PRINT L;
   ID LIKFULL;
   TITLE4 "&COMP ";
RUN;
   TITLE4;
%MEND LR;
*********************************************************;
DATA NEW;
   INPUT TRT STIME CEN LOGWBC @@;
   TRT_LWBC = TRT*LOGWBC;
   CARDS;
1 6 0 3.20 1 6 1 2.31 1 6 1 4.06 1 6 1 3.28 1 7 1 4.43 1 9 0 2.80
1 10 0 2.70 1 10 1 2.96 1 11 0 2.60 1 13 1 2.88 1 16 1 3.60
1 17 0 2.16 1 19 0 2.05 1 20 0 2.01 1 22 1 2.32 1 23 1 2.57
1 25 0 1.78 1 32 0 2.20 1 32 0 2.53 1 34 0 1.47 1 35 0 1.45
0 1 1 2.80 0 1 1 5.00 0 2 1 4.91 0 2 1 4.48 0 3 1 4.01 0 4 1 4.36
0 4 1 2.42 0 5 1 3.49 0 5 1 3.97 0 8 1 3.52 0 8 1 3.05 0 8 1 2.32
0 8 1 3.26 0 11 1 3.49 0 11 1 2.12 0 12 1 1.50 0 12 1 3.06
0 15 1 2.30 0 17 1 2.95 0 22 1 2.73 0 23 1 1.97
;;;;
RUN;
TITLE 'ANDERSON, ET ALL, COX PROPORTIONAL HAZARDS MODEL '
'(PAGES 202-219)';
RUN;
   /* EXAMPLE 1 */
%LR(DAT=NEW,FULLIND=TRT LOGWBC TRT_LWBC,DFFULL=3,
REDIND=TRT LOGWBC,DFRED=2,PRINTF=N,PRINTR=N,
COMP=TEST FOR TREATMENT-BY-LOGWBC INTERACTION)
   /* EXAMPLE 2 */
%LR(DAT=NEW,FULLIND=TRT LOGWBC,DFFULL=2,
REDIND=LOGWBC,DFRED=1,PRINTF=N,PRINTR=N,
COMP=TEST FOR TREATMENT GROUP DIFFERENCES AFTER ADJUSTING FOR
LOGWBC )
   /* EXAMPLE 3 */
%LR(DAT=NEW,FULLIND=LOGWBC TRT_LWBC,DFFULL=2,
REDIND=LOGWBC,DFRED=1,PRINTF=N,PRINTR=N,
COMP=TEST FOR TREATMENT-BY-LOGWBC INTERACTION,STRATAV=TRT)
```